Calcolo della matrice dell'operatore lineare ``derivata'' nello spazio lineare dei polinomi di grado al piu 3

Enrico Bertolazzi
Esercitazioni Maple di Geometria

```
> # definisco una base per lo spazio dei polinomi

  p0 := 1 ;
  p1 := 1+x ;
  p2 := 1+x+x*x ;
  p3 := 1+x+x*x*x ;
```

$$p0 := 1 \tag{1}$$
$$p1 := 1 + x$$
$$p2 := 1 + x + x^2$$
$$p3 := 1 + x + x^3$$

```
> # scrivo un generico polinomio come combinazione
  # lineare degli elementi della base

  p := a*p0 + b * p1 + c * p2 + d * p3 ;
```

$$p := a + b\,(1 + x) + c\,\left(1 + x + x^2\right) + d\,\left(1 + x + x^3\right) \tag{2}$$

```
> # DP e` la mappa che dato il generico polinomio
  # p ne calcolo la derivata

  DP := (p) -> diff(p,x) ;
```

$$DP := p \rightarrow \frac{\partial}{\partial x}\,p \tag{3}$$

```
> # scrivo un generico polinomio P come combinazione
  # lineare degli elementi della base

  P := A*p0+B*p1+C*p2+D*p3 ;
```

$$P := A + B\,(1 + x) + C\,\left(1 + x + x^2\right) + \mathrm{D}\,\left(1 + x + x^3\right) \tag{4}$$

```
> # calcolo i coefficienti del polinomio P in modo che P = DP(p).
  # in questo modo ottendo delle relazioni lineari tra le coordinate
  # del polinomio p e P=DP(p) cioe` ottengo la matrice della
  # trasformazione lineare

  res := solve( {coeff(P,x,0)=coeff(DP(p),x,0),
                 coeff(P,x,1)=coeff(DP(p),x,1),
                 coeff(P,x,2)=coeff(DP(p),x,2),
                 coeff(P,x,3)=coeff(DP(p),x,3)}, {A,B,C,D} ) ;
```

(5)

$$res := \{D = 0, C = 3\,d, A = d - c + b, B = -3\,d + 2\,c\} \tag{5}$$

```
> # scrivo la trasformazione componente per componente
  L := subs(res,<A,B,C,D>) ;
```

$$L := \begin{bmatrix} d - c + b \\ -3\,d + 2\,c \\ 3\,d \\ 0 \end{bmatrix} \tag{6}$$

```
> # la matrice risulta essere quindi

  mat := <subs(a=1,b=0,c=0,d=0,L)|
          subs(a=0,b=1,c=0,d=0,L)|
          subs(a=0,b=0,c=1,d=0,L)|
          subs(a=0,b=0,c=0,d=1,L)> ;
```

$$mat := \begin{bmatrix} 0 & 1 & -1 & 1 \\ 0 & 0 & 2 & -3 \\ 0 & 0 & 0 & 3 \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{7}$$